

```

program ::= class id { variable_declarations method_declarations }
variable_declarations ::= type variable_list ; variable_declarations |  $\epsilon$ 
type ::= int | real
variable_list ::= variable more_variables
more_variables ::= , variable_list |  $\epsilon$ 
variable ::= id opt_array
opt_array ::= [ num ] |  $\epsilon$ 
method_declarations ::= method_declaration more_method_declarations
more_method_declarations ::= method_declaration more_method_declarations |  $\epsilon$ 
method_declaration ::= static method_return_type id ( parameters )
                        { variable_declarations statement_list }
method_return_type ::= type | void
parameters ::= parameter more_parameters |  $\epsilon$ 
more_parameters ::= , parameters |  $\epsilon$ 
parameter ::= type id
statement_list ::= statement statement_list |  $\epsilon$ 

statement ::=      id assign_indec_func_call ;
                  | if ( expression ) statement_block optional_else
                  | for ( variable_loc = expression ; expression ; incr_decr_var ) statement_block
                  | return optional_expression ;
                  | break ;
                  | continue ;
                  | statement_block

assign_indec_func_call ::= opt_index assign_or_inc
                          | ( expression_list )

assign_or_inc ::= incdecop | = expression

optional_expression ::= expression |  $\epsilon$ 
statement_block ::= { statement_list }
incr_decr_var ::= variable_loc incdecop
optional_else ::= else statement_block |  $\epsilon$ 
expression_list ::= expression more_expressions |  $\epsilon$ 
more_expressions ::= , expression more_expressions |  $\epsilon$ 
expression ::= simple_expression optional_relop
optional_relop ::= relop simple_expression |  $\epsilon$ 
simple_expression ::= sign term optional_addops | term optional_addops
optional_addops ::= addop term optional_addops |  $\epsilon$ 
term ::= factor optional_mulop
optional_mulop ::= mulop term |  $\epsilon$ 
factor ::= id opt_array_func_call | num | ( expression ) | ! factor
opt_array_func_call ::= opt_index | ( expression_list )
variable_loc ::= id opt_index
opt_index ::= [ expression ] |  $\epsilon$ 
sign ::= + | -

```